

## drawdot in MetaPost: A bug, a fix

Hans Hagen


It is no secret that Don Knuth uses MetaPost for graphics in his books nowadays. This has the nice side effect of large-scale testing of MetaPost stability. Recently he uncovered a bug in the `drawdot` macro, which plain MetaPost has always defined like this:

```
def drawdot expr z =      % original definition
  addto currentpicture
  contour makepath currentpen shifted z _op_
enddef;
```

The submitted test was this:

```
for j = 1 upto 9 :
  pickup pencircle scaled .4;
  drawdot (10j,0) withpen pencircle scaled .5j;
  pickup pencircle scaled .5j;
  drawdot (10j,10);
endfor;
```


which visualizes as:



Let's simplify and exaggerate this a bit:

```
drawdot origin withpen pencircle scaled 2cm;
pickup pencircle scaled 2cm;
drawdot origin shifted (3cm,0);
```

The left-hand variant demonstrates that the old definition of the macro uses the current pen (which by default is one base unit,  $\frac{1}{65536}$  of a pixel) to calculate a contour (a.k.a. outline) that then is drawn with a larger pen. The opened up dot is a side effect of the exported PostScript code. The right-hand version shows that picking up the larger pen first and then drawing has a different (and correct) effect.




The two formulations should be equivalent. So the version of MetaPost that will ship with T<sub>E</sub>X Live 2014 has a new definition of `drawdot`. While the original definitions followed a METAFONT approach, the new definition relies on PostScript doing the work:

```
def drawdot expr p =      % new definition
  if pair p :
    addto currentpicture
    doublepath p withpen currentpen _op_
  else :
    errmessage("drawdot needs a pair ...")
  fi
enddef;
```

```
drawdot origin withpen pencircle scaled 2cm;
pickup pencircle scaled 2cm;
drawdot origin shifted (3cm,0);
```

Now our simplified example comes out the same:



This definition is more or less the same as:

```
let drawdot = draw;
```

But our more extensive variant has the advantage that it behaves a bit like a primitive operation: a dot is supposed to be a pair and if not, we get an error.

We believe that most users will not notice this change. First of all we have never received a complaint before, which might be an indication that users already used `draw` instead of `drawdot`. Second, dots are normally drawn small, so users might never have noticed such artifacts.

Of course the MetaPost team is curious about what bug Don will come up with next, especially when he needs very large graphics that rely on the new `double` (floating-point) mode of MetaPost.

In the original message, available at [tug.org/pipermail/tex-k/2014-January.txt](http://tug.org/pipermail/tex-k/2014-January.txt), a few more observations were made and testing revealed that there is room for improvement for paths that consist of a point cycling to itself. We will look into these some time in the future.

◇ Hans Hagen  
 Pragma ADE  
<http://metapost.org>

Editor's note: The figures here are bitmaps, extracted from screenshots, because the conversion of EPS to PDF for production also filled in the open dots! Clearly the effects are dependent on the particular software involved.